Point Charge Between Two Parallel Conducting Plates

PROBLEM:

A point charge Q is located between two infinite grounded parallel conducting sheets at a distance of d_y from the plate passing through the origin, as shown (note the coordinate system used including the position of the origin O).

The separation between the plates = L. We wish to find the electrostatic potential V[x,y,z] for points between the plates.

Single Point Charge Q positioned between two infinite, grounded, parallel conducting sheets



(a) What classic technique or method that is used for determining V and E for charges near certain symmetric geometries of conductors is your likely choice??

<Enter your answer in this text cell>

The method of images. Our overall goal is to determine the magnitude, sign, and position of image charges that will satisfy the boundary conditions, namely that V on the two parallel planes is zero, and then generate V (and E) in the region x >0; y > 0.

(b) In this type of problem, what role do boundary conditions play in finding V and E?

<Enter your answer in this text cell>

The uniqueness of the solution to Laplace's Equation (LE) for the electric potential in a region of space arises when (say) a general solution is "hammered" into satisfying the boundary conditions (e.g., potentials on surrounding surfaces). Generally this results in constants of integration taking on specific, unique values. Turning it around, if we find a solution (by hook or by crook) to LE, it is unique and we are done.

It is understood that the field of a point charge $\left(\sim \frac{1}{r}\right)$ satisfies LE. We can quickly use M to verify this by using the Laplacian Function in Spherical Coordinates and stick in V(r) ~ $\frac{1}{r}$. If we get 0, we've Verified (Not Proved!).

In[171]:=
```
(* Execute code below *)

ClearAll["`*"]
Laplacian[1/r, {r, θ, ϕ}, "Spherical"]
```

Out[172]= 0

So above, *Your Response* included *Images charges??* Brilliant! The challenge is to find a set of point charges that will produce a zero-potential on both conducting planes. Hints:

We use the results of using a single image charge for the case of a single real charge Q a distance $+d_y$ in front of an infinite grounded conducting (x-z) plane. The correct image charge in this example, Q_{image}= -Q, is placed a distance d_y *behind* the plane. This will yield a zero potential along the entire x-z plane located at y = 0.

Single Plane; point charge:



Reminder: To find V[x,y,z] for y > 0 ONLY, we take simply add $V_{(Q)} + V_{(-Q)}$;

$$\text{i.e.,}\quad V[x,y,z] = k\,Q\left(\frac{1}{\sqrt{x^2+(y-d_y)^2+z^2}} + \frac{1}{\sqrt{x^2+(y+d_y)^2+z^2}}\right)$$

(c) Determine the magnitudes and positions of the image charges such that both planes are at zero volts potential [Superhint: image charges will all be located along the ± y axis.]

<Enter your answer in this text cell>

We first must recognize that a single image charge is not sufficient for the two plane problem. In fact we are going to need an INFINITE number of charges of magnitudes ± Q located at increasing distances from O along the y axis.

We start with placing the first image charge of magnitude -Q at the position {0, -dy, 0}. This insures a zero potential on the right hand plane (at y = 0), just like in the single plane example.

BUT NOW the second plane (at y = L) is no longer an equipotential (due to the TWO charges: +Q at {0, dy, 0} and -Q at {0, –dy, 0}).

We respond aggressively! by placing two additional image charges, –Q at {0, 2L–dy, 0} and a +Q at {0, 2L+dy, 0} which now makes the y = L plane an equipotential.

BUT NOW the first plane at y = 0 no longer an equipotential (due to adding these last two image charges).

So again we respond aggressively! by placing two additional image charges, Q at {0, –2L+dy, 0} and a –Q at {0, –2L–dy, 0} which now makes the y = 0 plane an equipotential.

You get the point — it's turtles all the way. We keep adding (aggressively!) pairs of ±Q image charges at distances further and further away [in the ±y directions] from the two plates. Here is the 2D picture of the charges and locations in the x = 0 plane for a few "cycles". The d_y's are the increments of length from the vertical planes shown:



As you might expect, the farther the image charges are from the conducting planes, the less they will contribute to the potential in the space between the plates, so in practice we'll be able to terminate the "series".

Staring at the above figure for several hours we can surmise that charges are located as follows (where dy > 0):

(i) –Q image charges at the following values of y: (2 n L – dy) for all integers including zero

(ii) +Q image charges at the following values of y: (2 n L + dy) for all integers excluding zero.

(iii) The original REAL charge, +Q, at y = dy.

NOTE that if we combine (b) and (c) we will get all +Q positions for the following values of y (2 n L + dy) all integers including zero.

If we use the notation for positions of source charges (the rprimes) and the field point (r), we can write the potentials due to these charges in terms of the corresponding (ℝ = r - rprime) values. (ℝ is what is often called the separation vector -- the displacement vector from a charge element to a field point.)

(ii) rprimenegative[n_] = {0, 2 n L – dy, 0}

(iii) rprimepositive[n_] = {0, 2 n L + dy, 0}

The position of the field point is r = {x,y,z} where -∞ < x < ∞; 0 < y < L; –∞ < z < ∞.

The ℝ vectors are of the form: r – rprime.

The potential due to any one of these charges is of the form: $\frac{k\,Q}{\text{Length of the appropriate ℝ vector}}$

Here's the WHOLE DEAL for V[x,y,z]: V[x,y,z] =

$$\sum_{n=-nmax}^{nmax} \frac{+k\,Q}{\sqrt{\text{ℝpositive}[n].\text{ℝpositive}[n]}} + \sum_{n=-nmax}^{nmax} \frac{-k\,Q}{\sqrt{\text{ℝnegative}[n].\text{ℝnegative}[n]}}$$

(d) Stick your results into M; use ± nmax for the limits on the sums over n instead of ± ∞.

In[182]:=

```
(* Input code below *)
Evaluation → QuitKernal → Local; (*  this does a super
 scrub of constants and variables that might be problematic  *)
ClearAll["Global`*"]

r = {x, y, z};
rprimenegative[n_] = {0, 2 n L - dy, 0};
(*  source points for the negative image charges  *)
rprimepositive[n_] = {0, 2 n L + dy, 0};
(*  source points for the positive image charges AND the real +
 Q charge at {0,dy,0}  *)

ℝpositive [n_] = r - rprimepositive[n];
ℝnegative [n_] = r - rprimenegative[n];
V[x_, y_, z_] =
```

$$\sum_{n=-nmax}^{nmax} \frac{+k\,Q}{\sqrt{ℝpositive[n].ℝpositive[n]}} + \sum_{n=-nmax}^{nmax} \frac{-k\,Q}{\sqrt{ℝnegative[n].ℝnegative[n]}};$$

(e) Put in values for the constants. I used: k = 1; Q = 1; L = 2; dy = 0.3; nmax = 30. Make a 3D Plot of V in the z = 0 plane (i.e., of V[x,y,0]).

In[190]:=
```
(* Input code below *)
k = 1; Q = 1; L = 2; dy = 0.3; nmax = 30;
Print[
 Style["This is a Plot of the Potential BETWEEN THE PLATES as a function of
    x,y in the z = 0 plane (which contains Q).  Use your
    mouse to rotate and/or resize:", Red, Bold, 16]]
Plot3D[V[x, y, 0], {x, -2 L, 2 L}, {y, 0, L}, PlotRange → {0, 5},
 BoxRatios → Automatic, MaxRecursion → 6,
 PlotLabel → Style[Framed["V[x,y,0] Between Plates"], 16, Blue, Bold,
   Background → Lighter[Orange]], AxesLabel → {"x", "y", "V[x,y,0]"}]
```

This is a Plot of the Potential BETWEEN THE PLATES as a function of x,y in the z = 0 plane (which contains Q). Use your mouse to rotate and/or resize:

Out[192]=



This is the V sought in the region where it is defined (between the plates) for the L, dy (0.3), and Q chosen. Using a finite nmax is, of course, an approximation. However, varying nmax to any value > ~ 8, it is very hard to see any visual change in this plot.

Now let's show the potential for both the REAL region between the plates and in the region where you're not supposed to look or go! (y < 0 and y > L). Then you see the regions where there are image charges.

Although not obvious, all of the image charges are contributing to V between the plates. I've outlined the "real region" with a black rectangle.

In[193]:=
```
(* Execute code below *)
Print[Style["This is a Plot of the Potential due to Q and a few of the
    image charges (again, staying in the z = 0 plane).   Only the
    outlined region (for 0 > y > L) between the plates is 'real'.
    Use your mouse to rotate and/or resize. ", Brown, Bold, 16]]
plotty = Plot3D[V[x, y, 0], {x, -2 L, 2 L}, {y, -4 L, 4 L}, PlotRange → {-5, 5},
   BoxRatios → {8, 16, 10}, MaxRecursion → 6,
   PlotLabel → Style[Framed["V[x,y,0] showing image charges"], 16, Blue, Bold,
     Background → Lighter[Orange]], AxesLabel → {"x", "y", "V[x,y,z]"}];
liny = Graphics3D[{Thick, Line[{{-2 L, 0, 0}, {2 L, 0, 0},
      {2 L, L, 0}, {-2 L, L, 0}, {-2 L, 0, 0}}]}];
Show[
 plotty,
 liny]
```

This is a Plot of the Potential due to Q and a few of the image charges (again, staying in the z = 0 plane).   Only the outlined region (for 0 > y > L) between the plates is 'real'. Use your mouse to rotate and/or resize.

Out[196]=



To repeat, in this plot, only the region outlined in black is in the space between the plates where V is actually defined/valid. Also, it is obvious that there are both + and – Q image

charges corresponding to the + and - peaks in V. For d_y such that Q is close to either plate, the system looks like a string of dipoles separated by a distance 2L.

(f) Before pursuing further interpretation of the V plot, I want to use Manipulate and show the above for different values of d_y (moving the charge from one plate towards the other). Run this cell and vary d_y by using the slider.

Note: When you change d_y, it may take some time to recompute V and update the plot. I also want to plot V of the point charge by itself, Vpoint[x,y,0], for comparison.

M Note: I used an option BoxRatios→Automatic which scales the axes (most importantly (x,y)) 1:1. This insures that the plot is not distorted.

Changing dy on one plot changes the other simultaneously so you can easily compare.

Use your mouse to rotate and modify the size of the plots for best viewing.

```
In[197]:= (* Execute code below *)
Clear[dy];
Vpoint[x_, y_, z_] = +k Q / Sqrt[ℝpositive[0].ℝpositive[0]];

Print[
 Style["This is a Plot of the desired Potential V[x,y,0] BETWEEN THE PLATES
    as a function of x,y in the z = 0 plane.  With the
    slider, we can Manipulate the distance (dy) of the
    point charge from the y = 0 plate.  ", Red, Bold, 16]]
Print[Style["      \nSUGGESTION: Rather than using the slider,
    IT'S BEST TO JUST ENTER A NEW dy IN THE WINDOW displaying
    the magnitude of dy and hit RETURN. ", Blue, Bold, 14]]
Print[Style["      (When you change dy, it will take some time
    to recompute V and update the plot)", Green, Bold, 14]]
Manipulate[dy; Plot3D[V[x, y, 0], {x, -2 L, 2 L}, {y, 0, L},
  PlotRange → {0, 5}, BoxRatios → Automatic, MaxRecursion → 6,
  PlotLabel → Style[Framed["V[x,y,0] Between Plates"], 16, Blue, Bold,
    Background → Lighter[Orange]], AxesLabel → {"x", "y", "V[x,y,0]"}],
 {dy, .1, 1.9, Appearance → "Open"}, LocalizeVariables → False]

Print[Style[
  "Below is a Plot of the Potential of the Single POINT CHARGE Q only BETWEEN
    THE PLATES (no image charges) as a function of x,y in the z = 0 plane
    where we can Manipulate the distance (dy) of the point charge from
    the y = 0 plate.  NOTE:  Changing dy on either of these Manipulate
    plots CHANGES BOTH -- this is an M thingy. ", Red, Bold, 16]]


Manipulate[dy; Plot3D[Vpoint[x, y, 0], {x, -2 L, 2 L}, {y, 0, L},
  PlotRange → {0, 5}, BoxRatios → Automatic, MaxRecursion → 6,
  PlotLabel → Style[Framed["Vpoint[x,y,0] Between Plates"], 16, Blue, Bold,
    Background → Lighter[Orange]], AxesLabel → {"x", "y", "V[x,y,0]"}],
 {dy, .1, 1.9, Appearance → "Open"}, LocalizeVariables → False]
```

This is a Plot of the desired Potential V[x,y,0] BETWEEN THE PLATES as a function of x,y in the z = 0 plane. With the slider, we can Manipulate the distance (d_y) of the point charge from the y = 0 plate.

SUGGESTION: Rather than using the slider, IT'S BEST TO JUST ENTER A NEW d_y IN THE WINDOW displaying the magnitude of d_y and hit RETURN.

(When you change d_y, it will take some time to recompute V and update the plot)

Out[202]=



Below is a Plot of the Potential of the Single POINT CHARGE Q *only* BETWEEN THE PLATES (no image charges) as a function of x,y in the z = 0 plane where we can Manipulate the distance (d_y) of the point charge from the y = 0 plate. NOTE: Changing dy on either of these Manipulate plots CHANGES BOTH -- this is an M thingy.

Out[204]=



(g) Now interpret these results (e.g., where are the plates? Shape of plots; compare?? Putting charge in center vs. very close to plates? Enter in the text cell below:

<Enter interpretation in this text cell>

The plates are located along the planes associated with y = 0 and y = L. Remember that the values of the scalar FUNCTIONS V[x, y, 0] and Vpoint[x, y, 0] are what is being plotted in the vertical direction. ALSO, remember that we have set z = 0 in these functions. We are exploring the V's between the plates in the same plane that hosts the real point charge Q.

In both plots, the infinity at the position of the real point charges {0, dy, 0} is clearly observed. when dy = L/2, both plots show the expected symmetry (you could call it the lack of ϕ dependence). Vpoint maintains that symmetry for all dy because it's simply the V of a single point charge.

V[x,y,0] for dy very small shows the ESSENTIAL fall of V to zero all along the grounded plate at y = 0 (and at the plate located at y = L). The point charge alone has no idea the grounded plate exists so exhibits the 1/r fall off as expected. Clearly, Vpoint does not fall to zero at the plates.

Same for dy very close to L near the other plate.

IF you wanted to explore V and Vpoint for other values of z (above or below the x-y plane), say zo, you can change the V[x, y, 0] and Vpoint[x, y, 0] to V[x, y, zo] and Vpoint[x, y, zo]. One result will be that the infinities in the V's will vanish (i.e., V[x, dy, zo] and Vpoint[x, dy, zo] are finite. Why? Because for zo ≠ 0, $\sqrt{\mathbb{R}\text{positive}[n] . \mathbb{R}\text{positive}[n]}$ and $\sqrt{\mathbb{R}\text{negative}[n] . \mathbb{R}\text{negative}[n]}$ do not go to zero. Thus, their reciprocals are finite.)

CONTOURS OF V:

A "vanilla" contour plot of V in the z = 0 plane looks like this:

In[205]:=
```
(* Execute code below *)
dy = 0.3;
contourplot = ContourPlot[V[x, y, 0], {x, -2 L, 2 L}, {y, 0, L},
  AspectRatio → Automatic, Frame → True, FrameLabel → {"x axis", "y axis"}]
```

Out[206]=



M blindly determines the contours it will display; to display more contours we have to play around. I have kluged a set of contours (values of V) using a bunch of Tables Joined together ("cons" is a list of these values). Once executed, you can point your mouse at any contour on the plot and read its value.

In[207]:=
```
(* Execute code below *)
cons = Join[Table[20 - i, {i, 0, 20}], Table[1 - 0.1 i, {i, 0, 9}],
  Table[0.1 - 0.01 i, {i, 0, 9}], Table[0.01 - 0.001 i, {i, 0, 9}]]

Print[
 Style["Here is a ContourPlot of the Potential BETWEEN THE PLATES as a function
    of x,y IN THE z = 0 plane (which contains Q):", Purple, Bold, 16]]
contoury = contourplot = ContourPlot[V[x, y, 0], {x, -2 L, 2 L}, {y, 0, L},
   PlotRange → {0, 10}, ColorFunction → Hue, Contours → cons, Frame → True,
   FrameLabel → {"x ", "y "}, PlotLabel → "V[x,y,0] Contours",
   MaxRecursion → 2, AspectRatio → Automatic, ImageSize → 500]
```

Out[207]=
```
{20, 19, 18, 17, 16, 15, 14, 13, 12, 11, 10, 9, 8, 7, 6, 5, 4, 3, 2, 1, 0, 1., 0.9, 0.8,
 0.7, 0.6, 0.5, 0.4, 0.3, 0.2, 0.1, 0.1, 0.09, 0.08, 0.07, 0.06, 0.05, 0.04, 0.03,
 0.02, 0.01, 0.01, 0.009, 0.008, 0.007, 0.006, 0.005, 0.004, 0.003, 0.002, 0.001}
```

Here is a ContourPlot of the Potential BETWEEN THE PLATES as a function of x,y IN THE z = 0 plane (which contains Q):

Out[209]=



(h) Now find E (suggest you use EE) in this segment. (Hint: use –Grad in Cartesian Coordinates). Remember you have to have nnmax defined; I keep nnmax = 30.

In[210]:=
```
(* Input code below *)
Clear[k , Q , L, dy]; (* Leave the Clear command *)

nmax = 30;
EE[x_, y_, z_] = - Grad[V[x, y, z], {x, y, z}];
(* if you really want to see it, remove the ; *)
```

(i) Now we want to do some plotting of the field, comparing with the above ContourPlot, and a look at the induced charges on the plates. Let me walk you through this.

First, plotting the vector E field is challenging (to see all of it's features). First, let's show a StreamPlot to get a general feel for the field. Here, I put it on top of the ContourPlot above (I've stayed with dy = 0.3). This code refers to EE[x_,y_.z_], which you hopefully defined in Part (i).

In[213]:=
```
(* Execute code below *)
k = 1; Q = 1; L = 2; dy = 0.3;
EEE[x_, y_] = { EE[x, y, 0][[1]], EE[x, y, 0][[2]]};
streamy = StreamPlot[EEE[x, y], {x, -2 L, 2 L}, {y, 0, L},
   AspectRatio → Automatic, ImageSize → 500, StreamColorFunction → "Rainbow"];
Show[contoury, streamy]
```

Out[216]=



The Streamlines are curves that trace tangents to the E field which (symbolically) indicate the flow of the field (think fluids — i.e., the flow velocity field). Magnitudes of E are not shown.

The plot above shows that the directions of E are normal to the contour lines, consistent with the directions of maximum gradient (E = – Grad V). Also note that the streamlines are approaching each plate (at y = 0 and y = L) in a normal direction. The plates are both equipotentials (0 V) and so E should be normal to the plates. Note also that the directions of E is consistent with negative charge on both plates (induced by the presence of +Q); i.e., the field lines originate on + Q and terminate on the negative charges distributed on the two plates.

Somewhat misleading is the missing evidence of E Field directly in the path between +Q (at {0, dy} and {0,0}. The contours are changing in large steps of V and are closely spaced (decreasing as y -> 0). E should be BIG.

Again, it is challenging to show a VectorPlot of V between the plates. Here is the Vanilla Version (it shows One Big Vector!):

In[217]:=
```
(* Execute code below *)
VectorPlot[EEE[x, y], {x, -2 L, 2 L}, {y, 0, L},
 AspectRatio → Automatic, Frame → True, FrameLabel → {"x ", "y "}]
```

Out[217]=



Not much better, but with some tricks I can milk out a little more of the field:

In[218]:=
```
(* Execute code below *)
EEEE[x_, y_] = If[Norm[EEE[x, y]] > 50, {0, 0}, EEE[x, y]];
vecy = VectorPlot[EEEE[x, y], {x, -2 L, 2 L}, {y, 0, L},
  AspectRatio → Automatic, ImageSize → 500, Frame → True, FrameLabel → {"x ", "y"}]
```

Out[219]=



The bottom line, is the field is indeed the largest in the line between the location of the +Q and the closest point on the bottom plate {0,0} and falls off rapidly in all other directions.

Let’s simply plot the amplitude of the (x and y) components of he E field in the z = 0 plane along the path from the origin {0,0} [in the center of the x = 0 plate] to the point {0, L} [in the center of the y = L plate]; the arrow shows the y position of the +Q.

In[220]:=
```
(* Execute code below *)
dy = 0.3;
dypos = Graphics[{Arrow[{{dy, 1600}, {dy, 100}}], Text["dy", {dy, 1800}]}];
plotEamplitudes =
  Plot[{EE[0, y, 0][[1]], EE[0, y, 0][[2]]}, {y, 0, L}, PlotRange → {-2000, 2000},
   AxesLabel → {"y", "EEx and EEy"} , PlotLegends → {"EEx", "EEy"}];
Show[plotEamplitudes, dypos ]
```

Out[223]=



You can see that EE_x is zero (the straight blue line along the y axis) for all y along this path (due to symmetry); same for EE_z, by the way. The switch in sign seen in EE_y is expected because of the switch in direction of EE at y = d_y where, of course, infinities on each side are observed.

Now look at the induced charge distributions on the two electrode surfaces. You recall that at a conducting surface, $\sigma = \frac{E_\perp^{\text{surface}}}{\epsilon o}$. The sign of σ is + if $E_\perp^{\text{surface}}$ points out of the surface and - if $E_\perp^{\text{surface}}$points towards the surface.

It's not hard to see that $E_\perp^{\text{surface}}$ at the y = 0 plane is EE[0,y][[2]] and $E_\perp^{\text{surface}}$ at the y = L plane is EE[0,L][[2]]. (The [[2]] selects the y component of EE)

So we can quickly generate the charge densities on the two planar electrodes in the problem (I call them sigmazero = "sigma_y = 0_plane" and sigmaL = "sigma_y = L_plane"). At both surfaces, σ is negative. I had to put in a – sign for sigmaL to force sigmaL to be negative. [This reflects that fact that the surface normal for the surface at y = 0 points to the right (into the region of interest), while the surface normal for the surface at y = 0 points to the left (also into the region of interest).]

In[224]:=
```
(* Execute code below *)
sigmazero = EE[x, 0, z][[2]] / ϵo ;
sigmaL  = - EE[x, L, z][[2]] / ϵo ;
 ϵo = 1;

Plot3D[sigmazero, {x, - 2 L, 2 L}, {z, -2 L, 2 L},
 PlotRange → {0, -20}, AxesLabel → {"x", "z", "sigma_y = 0_plane"}]
Plot3D[sigmaL, {x, - 2 L, 2 L}, {z, -2 L, 2 L}, PlotRange → {0, -1},
 AxesLabel → {"x", "z", "sigma_y = L_plane"}]
```

Out[227]= sigma_y = 0_plane



Out[228]= sigma_y = L_plane



Note that the sigma scales are different — with Q close to the y = 0 plane, the sigma on the y = L plane is considerably smaller in line with the much reduced E field at y = L relative to y = 0. For L = 2 and dy = 1 (Q right in the middle), here are the charge distributions. Use your mouse to tilt and rotate the plots.

In[229]:=
```
(* Execute code below *)
dy = 1;
sigmazero = EE[x, 0, z][[2]] / ϵo;
sigmaL = - EE[x, L, z][[2]] / ϵo;
ϵo = 1;

Plot3D[sigmazero, {x, - 2 L, 2 L}, {z, -2 L, 2 L},
 PlotRange → {0, -2}, AxesLabel → {"x", "z", "sigma_y = 0_plane"}]
Plot3D[sigmaL, {x, - 2 L, 2 L}, {z, -2 L, 2 L}, PlotRange → {0, -2},
 AxesLabel → {"x", "z", "sigma_y = L_plane"}]
```

Out[233]=



Out[234]=



As expected, they are the same due to the symmetric positioning of Q between the plates.

To rap it up, Q induces negative surface charge distributions (sigmazero and sigmaL) on the two plates. The field between the plates is determined by the sum of the fields due to these two charge distributions and the original +Q.

IF by magic we had known sigmazero and sigmaL [as functions of x,y,z] and the position of

+Q, we could, in principle, calculate EE between the plates. Certainly numerical calculations would be possible using the tools of Griffiths Chapter 2.

BUT we did not know sigmazero and sigmaL. Instead, we were able to use the Image Charge Method to calculate a Solution to Laplace's Equation (without solving it!) that satisfies the BCs. By *Uniqueness* we then have derived the desired V[x,y,z]. From this V, we calculated E[x,y,z] and the surface charge distributions on the plates.

Done.